

Contrasts and Multiple Comparisons

```
/* onewaymath.sas */
title2 'Oneway with contrasts and multiple comparisons (Exclude Other/DK)';
%include 'readmath.sas';
if ethnic ne 6; /* Otherwise, throw the case out */
proc freq;
  tables ethnic;
proc glm;
  title3 'Do average marks differ significantly by apparent nationality?';
  class ethnic;
  model grade = ethnic / clparm; /* clparm gives CI for contrasts down in
                                the estimate statement. */
  means ethnic;
  /* Test custom contrasts, or "planned comparisons"
     (Note Alphabetical Order)
     Asian
     East Indian
     Eastern European
     European not Eastern
     Middle-Eastern and Pakistani */
  contrast 'European_vs_Other' ethnic -2 -2 3 3 -2;
  estimate 'European_vs_Other' ethnic -2 -2 3 3 -2 / divisor=6;
  estimate "EastEurop_vs_OtherEurop" ethnic 0 0 .5 -.5 0;
  contrast 'Non-Europ_Equal?' ethnic 1 -1 0 0 0,
          ethnic 1 0 0 0 -1;
  contrast 'East Indian vs. Middle-Eastern and Pakistani'
          ethnic 0 1 0 0 -1;
  contrast 'East Indian vs. Eastern European'
          ethnic 0 1 -1 0 0;
  contrast 'ReplicateOverallF=2.45'
          ethnic 1 -1 0 0 0,
          ethnic 1 0 -1 0 0,
          ethnic 1 0 0 -1 0,
          ethnic 1 0 0 0 -1;
  /* Multiple Comparisons */
  lsmeans ethnic / pdiff adjust=bon;
  lsmeans ethnic / pdiff adjust=tukey;
  lsmeans ethnic / pdiff adjust=scheffe;
  /* Get Scheffe critical value from proc iml */
proc iml;
  title3 'Scheffe critical value for all possible contrasts';
  numdf = 4; /* Model degrees of freedom for initial test */
  dendf = 373; /* Error degrees of freedom for initial test */
  alpha = 0.05;
  critval = finv(1-alpha,numdf,dendf);
  scrit = critval * numdf;

  print "Initial test has" numdf " and " dendf "degrees of freedom."
        "-----"
        "Using significance level alpha = " alpha
        "-----"
        "Critical value for the initial test is " critval
        "-----"
        "Critical value for Scheffe tests is " scrit
        "-----";
```

Here is **onewaymath.lst**

Gender, Ethnicity and Math performance 1

The FREQ Procedure

Judged Nationality of name

ethnic	Frequency	Percent	Cumulative Frequency	Cumulative Percent
Asian	131	24.30	131	24.30
Eastern European	63	11.69	194	35.99
European not Eastern	195	36.18	389	72.17
Middle-Eastern and Pakistani	72	13.36	461	85.53
East Indian	78	14.47	539	100.00

Gender, Ethnicity and Math performance 2

Do average marks differ significantly by apparent nationality?

The GLM Procedure

Class Level Information

Class	Levels	Values
ethnic	5	Asian East Indian Eastern European European not Eastern Middle-Eastern and Pakistani
		Number of Observations Read 539
		Number of Observations Used 378

Gender, Ethnicity and Math performance 3

Do average marks differ significantly by apparent nationality?

The GLM Procedure

Dependent Variable: grade Final mark (if any)

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	4	3640.6607	910.1652	2.45	0.0458
Error	373	138545.4478	371.4355		
Corrected Total	377	142186.1085			

R-Square	Coeff Var	Root MSE	grade Mean
0.025605	32.80823	19.27266	58.74339

Source	DF	Type I SS	Mean Square	F Value	Pr > F
ethnic	4	3640.660655	910.165164	2.45	0.0458

Source	DF	Type III SS	Mean Square	F Value	Pr > F
ethnic	4	3640.660655	910.165164	2.45	0.0458

Gender, Ethnicity and Math performance

4

Do average marks differ significantly by apparent nationality?

The GLM Procedure

Level of ethnic	N	-----grade----- Mean	Std Dev
Asian	87	60.0574713	20.9314253
East Indian	53	65.1886792	18.5317364
Eastern European	46	55.7608696	20.2771736
European not Eastern	142	56.2816901	17.8581353
Middle-Eastern and Pakistani	50	59.3600000	19.9691190

lsmeans output appears next – skip for now and come back to it.

Do average marks differ significantly by apparent nationality?

The GLM Procedure
 Least Squares Means
 Adjustment for Multiple Comparisons: Bonferroni

ethnic	grade LSMEAN	LSMEAN Number
Asian	60.0574713	1
East Indian	65.1886792	2
Eastern European	55.7608696	3
European not Eastern	56.2816901	4
Middle-Eastern and Pakistani	59.3600000	5

Least Squares Means for effect ethnic
 Pr > |t| for H0: LSMean(i)=LSMean(j)

Dependent Variable: grade

i/j	1	2	3	4	5
1		1.0000	1.0000	1.0000	1.0000
2	1.0000		0.1567	0.0432	1.0000
3	1.0000	0.1567		1.0000	1.0000
4	1.0000	0.0432	1.0000		1.0000
5	1.0000	1.0000	1.0000	1.0000	

Deleting some output ...

Adjustment for Multiple Comparisons: Tukey-Kramer

i/j	1	2	3	4	5
1		0.5449	0.7380	0.6028	0.9996
2	0.5449		0.1103	0.0349	0.5410
3	0.7380	0.1103		0.9999	0.8914
4	0.6028	0.0349	0.9999		0.8680
5	0.9996	0.5410	0.8914	0.8680	

Adjustment for Multiple Comparisons: Scheffe

i/j	1	2	3	4	5
1		0.6747	0.8273	0.7228	0.9998
2	0.6747		0.2096	0.0854	0.6713
3	0.8273	0.2096		0.9999	0.9334
4	0.7228	0.0854	0.9999		0.9181
5	0.9998	0.6713	0.9334	0.9181	

Do average marks differ significantly by apparent nationality?

The GLM Procedure

Dependent Variable: grade Final mark (if any)

Contrast	DF	Contrast SS
European_vs_Other	1	2377.061841
Non-Europ_Equal?	2	1123.948886
East Indian vs. Middle-Eastern and Pakistani	1	874.075530
East Indian vs. Eastern European	1	2188.870763
ReplicateOverallF=2.45	4	3640.660655

Contrast	Mean Square	F Value	Pr > F
European_vs_Other	2377.061841	6.40	0.0118
Non-Europ_Equal?	561.974443	1.51	0.2216
East Indian vs. Middle-Eastern and Pakistani	874.075530	2.35	0.1259
East Indian vs. Eastern European	2188.870763	5.89	0.0157
ReplicateOverallF=2.45	910.165164	2.45	0.0458

Parameter	Estimate	Standard Error	t Value	Pr > t
European_vs_Other	-5.51410365	2.17969837	-2.53	0.0118
EastEurop_vs_OtherEurop	-0.26041029	1.63481102	-0.16	0.8735

Parameter	95% Confidence Limits	
European_vs_Other	-9.80014111	-1.22806619
EastEurop_vs_OtherEurop	-3.47501161	2.95419103

Scheffe critical value for all possible contrasts

numdf dendf
 Initial test has 4 and 373 degrees of freedom.

 alpha

Using significance level alpha = 0.05

 critval

Critical value for the initial test is 2.3958734

 scrit

Critical value for Scheffe tests is 9.5834935